

# Developing mixed Java/Groovy projects with Gradle using Netbeans

Kostas Saidis



Netbeans Day Athens 2016  
August 26, 2016

# About Me

Kostas Saidis

[saiko@niovity.com](mailto:saiko@niovity.com)

Software & Data  
Architect



## Academia

- ▶ BSc @ cs.unipi.gr (2001)
- ▶ MSc @ di.uoa.gr (2004)
- ▶ PhD @ di.uoa.gr (2011)

## Industry

- ▶ Freelancer since 1999 (consultant, developer & instructor)
- ▶ Entrepreneur since 2011 (niovity)

## Java

- ▶ Early Java enthusiast (1997)
- ▶ Groovyist since 2011

Nioivity



nioivity.com

**We are a software company**  
using a Java/Groovy/Javascript stack

# Nioivity Butterfly



**We offer a web-based repository platform  
for end-to-end data management**

Our clients include Academic Libraries, Public Benefit Foundations, Research Institutes and Private Companies

OpenDataCloud
API
Πολύγηση
API
Πληροφορίες
Επικοινωνία

# Αξιοποιώντας τα Ανοικτά Δημόσια Δεδομένα

Το OpenDataCloud είναι μια υπηρεσία νέφους για την αξιοποίηση των ανοικτών δεδομένων του Ελληνικού Δημοσίου.

[Follow @opendatacloudgr](#)
[Tweet](#)

## Για το Κοινό

Προσηλγήστε στα δεδομένα, φιλτράρετέ τα και δείτε τα σχετικά γραφήματα.

[Περισσότερα](#)

## Για τους Δημόσιους Φορείς

Αναδείξτε τα δεδομένα της υπηρεσίας σας και υποστηρίξτε τους στόχους της Ανοικτής Διακυβέρνησης.

[Περισσότερα](#)

## Για τους Προγραμματιστές

Αναπτύξτε εφαρμογές εύκολα και γρήγορα με το ισχυρό OpenDataCloud API.

[Περισσότερα](#)

### Παραδείγματα Γραφημάτων

☒ Grouped
☐ Stacked

☒ Μαθητές-Δημόσιο
☐ Μητρώμολοι-Δημόσιο

☐ Μαθητές-Ιδιωτικό
☐ Μητρώμολοι-Ιδιωτικό

☒ Μαθητές-Δημόσιο
☐ Μαθητές-Ιδιωτικό

2<sup>nd</sup> Prize

Public Open Data Hackathon 2014

## Target audience

- ☒ Java
- ☒ NetBeans
- ☐ Groovy
- ☐ Gradle

Motivation

2  
facts

# 1. The Java ecosystem is polyglot now

## The Java Platform

1. The Java Language
2. The Java Development Kit
3. The Java Virtual Machine (the key component)

## The Java Virtual Machine

- ▶ A rock-solid, enterprise-grade runtime environment that can execute multiple languages (not only Java).

## JVM Languages

- ▶ Groovy, Scala, Javascript, Ruby, Python, Clojure, Kotlin, Fantom and many others...



# Groovy

The swiss army knife of the Java developer

## 2. Building software sucks

### Software build process

- ▶ Develop
- ▶ Test
- ▶ Assemble
- ▶ Deploy
- ▶ Integrate

### Repeat

- ▶ Again and again and again

# Gradle

The finest build tool out there

# Workshop Structure

1. Introduce Groovy & Gradle
2. Build a mixed Java/Groovy Swing application using Gradle & Netbeans

# Workshop Goal



# Workshop Material and Setup Instructions

Workshop Web Page

[www.niovity.com/static/NetBeansDayAthens2016/](http://www.niovity.com/static/NetBeansDayAthens2016/)

## Meet Apache Groovy



# What is Groovy

A feature-rich, Java-friendly, multi-paradigm language for the JVM

- ▶ More than a decade-long history\*.
- ▶ Free and Open Source Software (Apache License v2).
- ▶ Joined the Apache Software Foundation in 2015.
- ▶ #16 in the TIOBE PL Popularity Index, August 2016<sup>†</sup>.

`groovy-lang.org`

 `@groovylang` `#groovylang`

---

\*Groovy was the first “second language” proposed for the JVM (back in 2004).

<sup>†</sup><http://www.tiobe.com/tiobe-index/>



## A companion language for Java

- ▶ Compiles into JVM bytecode and preserves Java semantics: seamless integration with Java.
- ▶ Provides a less verbose syntax.
- ▶ Augments Java with additional features (e.g. new methods in libraries).
- ▶ Dynamic in nature, yet optional typing and compile-time static features are also supported.
- ▶ Easy to learn and use.
- ▶ Supported by all major Java IDEs<sup>†</sup>.

---

<sup>†</sup>In NetBeans you should install the “All” download bundle, see [here](#).

# Why Groovy II

## Community and Ecosystem

- ▶ **Gradle**: Build tool.
- ▶ **Spock**: Testing and specification framework.
- ▶ **Grails**: Web development framework.
- ▶ **Griffon**: Desktop app framework.
- ▶ And many, many more<sup>§</sup>.

All existing Java tools, libraries and frameworks are directly reusable by Groovy.

---

<sup>§</sup>For an overview, check [here](#).


# Getting Started

## Groovy Tools

1. **groovyc**: Compiles groovy sources to JVM bytecode (class files).
2. **groovysh**: Executes code interactively (Read-Eval-Print Loop).
3. **groovyConsole**: GUI app for interactive code execution.
4. **groovy**: Executes groovy scripts (you can use it like bash, perl, python, etc).
5. **groovydoc**: Generates documentation (like javadoc).
6. **grape**: An embedded jar dependency manager.

# The best place to start

## Start from the Groovy console (GUI app)



```
1 list = [1, 2, 3]
2 println list.collect { i -> i + 1 }
```

```
groovy> list = [1, 2, 3]
groovy> println list.collect { i -> i + 1 }
```

```
[2, 3, 4]
```

Execution complete. Result was null. 2:36

Or the **Groovy Web Console**

# The Groovy Syntax

The majority of Java syntax is part of the Groovy syntax:

- ▶ packages
- ▶ imports
- ▶ control structures
- ▶ exception handling
- ▶ classes and methods
- ▶ object instantiation and method calls

**Some Gotchas:** Arrays, additional Groovy keywords, equals checks, etc.

# Default imports

The following imports are included by default:

- ▶ `java.lang.*`
- ▶ `java.util.*`
- ▶ `java.net.*`
- ▶ `groovy.lang.*`
- ▶ `groovy.util.*`
- ▶ `java.math.BigInteger` and `java.math.BigDecimal`

# Optionals

The following are optional:

- ▶ **semicolons:** required only when you write multiple statements in one line.
- ▶ **variable types:** you decide what should be dynamic (using `def`).
- ▶ **return statements:** the last evaluated expression is the default return value **Gotcha**.
- ▶ **parentheses:** in method/function invocations **Gotcha**.

# Groovy truth and equals

## Groovy truth

- ▶ Null value is false.
- ▶ Empty collection or map is false.
- ▶ Empty string is false.
- ▶ Zero numeric value is false.

## Gotcha

- ▶ The `==` operator performs value equality (like Java `equals()`).
- ▶ Use the `is` method for identity (available in every Groovy object).



## Examples

```
1 def f = null
2 def t = "string"
3 assert f == false
4 assert t == true
5 def list = []
6 def map = [a:1]
7 assert list == false
8 assert map == false
9 def s = ""
10 assert s == false
11 def i = 1
12 def z = 0.0
13 assert i == true
14 assert z == false
```

# Strings

Groovy supports:

- ▶ **Single quotes:** ordinary strings.
- ▶ **Double quotes:** ordinary strings with variable expansion (GStrings).
- ▶ **Triple quotes:** multi-line strings with variable expansion (GStrings).
- ▶ **Slashy strings:** strings enclosed in slashes; no need to escape backslashes (useful for regular expressions and file paths).
- ▶ operator overloading.

## Strings

```
1  def mlStr = """ I am a multi-line
2  string!"""
3  def name = 'Angus'; def surname = "Young"
4  //GStrings
5  def fullname = "$name $surname"
6  assert fullname == "Angus Young"
7  //operator overloading
8  def s = name * 3
9  assert s == "AngusAngusAngus"
10 s = fullname - name
11 assert s.trim() == surname
12 //slashy strings: preserve backslashes; no escaping is required
13 s = /\n\r/
14 assert s.size() == 4
```

# A better syntax for common things

## Collections & Maps

```
1 //Lists (ArrayList behind the scenes)
2 List<Integer> list = [1, 2, 3]
3 assert list[0] == 1
4 //operator overloading
5 list << 4
6 assert list.size() == 4
7 //Maps (LinkedHashMap behind the scenes)
8 Map<String, String> map = ['one': '1', "two": "2", three: '3']
9 assert list[1] == map['two'] as Integer
10 map.one = 'none'
11 //Java ordinary methods are here
12 assert map.get('one') == 'none'
13 list.add(4)
14 assert list == [1, 2, 3, 4, 4]
15 //Convert a List to a Set
16 Set s = list as Set
17 assert s == [1, 2, 3, 4]
```

# Additional Operators

- ▶ Safe navigation: `x?.method()`
- ▶ Elvis: `x = y ? : "no y"`
- ▶ Spread: `["java", "groovy"].size() → [4, 6]`
- ▶ and more... `<=>`, `=~`, `==~`, `.@`, `.&`

# Groovy is Java on Steroids!

## Groovy powerful switch statement

```
1  switch(val) {  
2      case "String":    //a string  
3          break  
4      case 10..100:     //a range  
5          break  
6      case Date:        //a date instance  
7          break  
8      case ~/gw+/:      //a reg-ex  
9          break  
10     case ['A', 'B']:  //a list  
11         break  
12     case { it instanceof Number && it > Integer.MAX_VALUE }  
13         //a closure  
14         break  
15     default:  
16         //the default, treated as an "else" in Groovy.  
17 }
```

## Groovy is Object-Oriented

- ▶ Supports Java Interfaces, Classes and Enums.
- ▶ Offers additional conventions and facilities (e.g. Groovy properties) that make our life easier.
- ▶ Supports Traits: a controlled way to implement multiple inheritance, avoiding the diamond issue.

# Everything is an object

## Example

```
1  def x = 1
2  int y = 1
3  assert x.class == Integer.class
4  assert y.class == Integer.class
5  def l = 1L
6  assert l.class == Long.class
7  def z = 0.3
8  assert z.class == BigDecimal.class
9  def flag = false
10 assert flag.class == Boolean.class
```



# Numbers

In Groovy all numbers are objects and BigDecimal arithmetic is used by default.

## Examples

```
1 def x = 3
2 assert x.plus(4) == x + 4
3 assert x.multiply(4) == x * 4
4 assert x.mod(4) == x % 4
5 //BigDecimal arithmetic
6 assert x/4 == x.div(4)
7 assert x/4 != x.intdiv(4)
8 assert 1/2 == 0.5
9 assert 1/3 == 0.3333333333
```

## Person.groovy

```
1 class Person {  
2     String name  
3     String surname  
4 }  
5 Person p = new Person(name:"N", surname:"S")  
6 assert p.name == "N"  
7 assert p.getName() == "N"  
8 test.surname = "F"  
9 test.setSurname("F")
```

# Closures

## Groovy supports functional programming through Closures

☞ A closure is an anonymous function together with a referencing environment.

Think of closures as anonymous blocks of code that:

- ▶ can accept parameters or return a value,
- ▶ can be assigned to variables,
- ▶ can be passed as arguments,
- ▶ capture the variables of their surrounding lexical scope.

# Example

```
1 //a closure assigned to a variable
2 def multiplier = { Number x, Number y -> x * y }
3 //invocation of the closure
4 assert multiplier(2, 3) == 6
5 //partial application
6 def doubler = multiplier.curry(2)
7 assert doubler(3) == 6
8 //another doubler
9 def otherDoubler = { it * 2 } //it -> the first arg
10 otherDoubler(3) == doubler(3)
```

# Collection Manipulation

```
1 //a groovy list (ArrayList behind the scenes)
2 def list = [12, 2, 34, 4, 15]
3 //Collection.collect(Closure c) - extra method available in
  Groovy JDK
4 //So, given a closure
5 def inc = { it + 1 }
6 //we can invoke
7 list.collect(inc) == [13, 3, 35, 5, 16]
8 //or pass the closure code directly
9 assert list.collect{ it + 1 } == [13, 3, 35, 5, 16]
10 //some more fun
11 list.findAll{ it > 10 }.groupBy{ it % 2 == 0 ? 'even' : 'odd' }
    == [even:[12, 34], odd:[15]]
```

# Owner and delegate

```
1 class Test {  
2     long x = 2  
3     def xTimes = { 1 -> x * 1 }  
4 }  
5 Test test = new Test()  
6 assert test.xTimes.owner == test  
7 assert test.xTimes.delegate == test  
8 test.xTimes(3) == 6  
9 test.x = 3  
10 test.xTimes(3) == 9  
11 def map = [x:4]  
12 assert test.xTimes.resolveStrategy == Closure.OWNER_FIRST  
13 test.xTimes.resolveStrategy = Closure.DELEGATE_FIRST  
14 test.xTimes.delegate = map  
15 assert test.xTimes(3) == 12
```

# Meta-programming

## Groovy supports meta-programming

- ▶ Categories, Annotations and AST transformations: compile-time metaprogramming.
- ▶ Expandos, Meta-classes, Meta-Object protocol: runtime metaprogramming.

**Groovy Builders:** a powerful concept based on closures and metaprogramming (covered later).

## Annotations example

```
1  @Canonical Person {  
2      String name  
3      String surname  
4  }  
5  //@EqualsAndHashCode, @ToString and @TupleConstructor  
6  def p1 = new Person(name:"N", surname:"S")  
7  //but also  
8  def p2 = new Person("N", "S")  
9  //Groovy == is Java's equals()  
10 assert p1 == p2
```



## Introduce a new method in Strings

```
1 String.metaClass.isUpperCase = {  
2     delegate.toCharArray().every{ Character.isUpperCase(it) }  
3 }  
4 String.metaClass.isLowerCase = {  
5     !delegate.isUpperCase()  
6 }  
7 assert "JAVA".isUpperCase()  
8 assert "groovy".isLowerCase()
```

# Dynamic and Static Features

- ▶ In Groovy, most of the type checking is performed at runtime.
- ▶ Groovy also supports duck typing.
- ▶ Use `@TypeChecked` & `@CompileStatic` annotations to maximize static checks during compilation.
- ▶ `@TypeChecked` performs static type checking, yet it dispatches methods through the MOP (permits runtime meta-programming).
- ▶ `@CompileStatic` = `@TypeChecked` + “static” method linking (no MOP); is the closest you can get to javac-generated bytecode (in terms of behavior & performance).

## Java

```
1 button.setOnClickListener(new View.OnClickListener() {  
2     @Override  
3     void onClick(View v) {  
4         startActivity(intent);  
5     }  
6 });
```

## Groovy

```
1 button.setOnClickListener = { startActivity(intent) }
```

## Example

```
1  class Test {  
2      def s1  
3      Integer s2 = 2016  
4      void test() {  
5          s1 = "NetBeansDay2016"  
6          assert s1.class == String.class  
7          s1 = 2016  
8          s1.class == Integer.class  
9          s2 = "NetBeansDay2016" //fails at runtime  
10     }  
11 }  
12 new Test().test()
```

# Enable static type checking

## Example

```
1  import groovy.transform.TypeChecked
2  @TypeChecked class Test {
3      def s1
4      Integer s2 = 2016
5      void test() {
6          s1 = "NetBeansDay2016"
7          assert s1.class == String.class
8          s1 = 2016
9          s1.class == Integer.class
10         s2 = "NetBeansDay2016" //fails at compile-time
11     }
12 }
13 new Test().test()
```

# Languages and Typing I

## A statically-typed language

Resolves the types of variables during compilation » you cannot change the type of a variable.

(Java, opt. Groovy)

## A dynamically-typed language

Resolves the types of variables at runtime » you can change the type of a variable.

(Ruby, opt. Groovy)

# Languages and Typing II

## A strongly-typed language

Guarantees type conformance » you can't coerce a variable to a wrong type.

(Java, Ruby, Groovy)

## A weakly-typed language

Has type abstractions that leak » you can screw everything up in all possible ways.

(C)

# Seamless Integration with Java

## As simple as possible

- ▶ You get all groovy magic by just adding a jar in the classpath.
- ▶ Call Groovy from Java == call Java from Java.
- ▶ Joint compilation: full mix of Java and Groovy.



## Fetcher.java

```
1 public interface Fetcher<V> {  
2     V fetch();  
3 }
```

## Person.groovy

```
1 @Canonical class Person implements Fetcher<String>{  
2     String name  
3     String surname  
4     @Override  
5     String toString() { "$surname, $name" }  
6     @Override  
7     String fetch() { toString() }  
8 }
```

## UsingPerson.java

```
1 public class UsingPerson {
2     public static void main(String[] args) {
3         Person p = new Person("Theodoros", "Kolokotronis");
4         System.out.println(p.fetch());
5     }
6 }
```

## Joint compilation

```
$ groovyc Fetcher.java Person.groovy UsingPerson.java -j
$ java -cp groovy-all.jar:. UsingPerson
$ Kolokotronis, Theodoros
```

Groovy can be used as a Java-powered scripting language

## Echo.java

```
1 public class Echo {  
2     public static void main(String[] args) {  
3         if (args != null && args.length > 0) {  
4             //for Java < 8 you need a third-party library  
5             System.out.println(String.join(" ", args));  
6         }  
7     }  
8 }
```

## Echo.groovy (Groovy script – no boilerplate)

```
1 if (args) {  
2     println args.join(' ')  
3 }
```

## Java is Groovy is Java

```
1 public class Echo {  
2     public static void main(String[] args) {  
3         if (args != null && args.length > 0) {  
4             //for Java < 8 you need a third-party library  
5             System.out.println(String.join(" ", args));  
6         }  
7     }  
8 }
```

```
$ mv Echo.java Echo.groovy
```

```
$ groovy Echo.groovy booh booh
```

```
$ booh booh
```

# Querying a MySQL database using JDBC

## db.groovy

```
1 @GrabConfig(systemClassLoader=true)
2 @Grab('mysql:mysql-connector-java:5.1.6')
3 import groovy.sql.Sql
4 try {
5     def driver = "com.mysql.jdbc.Driver"
6     def query  = "select count(*) as c, date(cDate) as d from table
7                  group by d order by c"
8     //Shoot it
9     def sql = Sql.newInstance(args[0], args[1], args[2], driver)
10    sql.eachRow(query) { row -> println "${row.c}:${row.d}" }
11 }
12 catch(e) { e.printStackTrace() }
```

# Exploit the full power of Java effectively

## Isn't this Groovy or what?

```
$ groovy db jdbc:mysql://localhost:3306/test test test123
```

### Output

```
1 9427:2004-08-20
2 6615:2004-10-29
3 5498:2004-10-08
4 5103:2004-08-31
5 4864:2004-10-14
6 4675:2004-10-31
7 4583:2004-10-05
8 4570:2004-08-21
9 4339:2004-09-30
10 4235:2004-10-30
```

## Meet Gradle



# What is Gradle

## Gradle is a software build tool

- ▶ Developed by Gradle Inc
- ▶ Open source: Apache License v2
- ▶ First release 2009, latest release August 2016 (3.0)
- ▶ Used by: Android, Spring IO, Linkedin, Netflix, Twitter and more...



# At a glance

## Gradle

- ▶ **Combines the best of Ant, Ivy and Maven**
  - ▶ Cross-platform file management
  - ▶ Dependency management
  - ▶ Conventions
- ▶ **In a smart and extensible way**
  - ▶ Deep API
  - ▶ Groovy (and Kotlin in 3.0) DSLs
  - ▶ Easy-to-write plugins
- ▶ **Ultimately offering**
  - ▶ Clean and elegant builds
  - ▶ Plenty of new features
  - ▶ Better build management

# Features

## Non exhaustive list

- ▶ Full-fledged programmability
- ▶ Both declarative and imperative
- ▶ Convention over configuration
- ▶ Transitive dependency management
- ▶ Multi-project builds
- ▶ Polyglot (not only Java)
- ▶ Numerous plugins and integrations
- ▶ Incremental and parallel execution
- ▶ Reporting and analytics
- ▶ Embeddable
- ▶ Great documentation

# A simple example

## A simple Java project

```
1  apply plugin: "java"
2  group      = "org.foo.something"
3  version    = "1.0-SNAPSHOT"
4  repositories {
5      mavenCentral()
6  }
7  dependencies {
8      compile      "commons-io:commons-io:2.4"
9      testCompile  "junit:junit:4.11"
10     runtime      files("lib/foo.jar", "lib/bar.jar")
11 }
```

The contents of `build.gradle` (aka the build script), placed in the root folder of the project.

# Core Concepts

**Build script:** a build configuration script supporting one or more projects.

**Project:** a component that needs to be built. It is made up of one or more tasks.

**Task:** a distinct step required to perform the build. Each task/step is atomic (either succeeds or fails).

**Publication:** the artifact produced by the build process.

# Dependency Resolution

**Dependencies:** tasks and projects depending on each other (internal) or on third-party artifacts (external).

**Transitive dependencies:** the dependencies of a project may themselves have dependencies.

**Repositories:** the “places” that hold external dependencies (Maven/Ivy repos, local folders).

**DAG:** the directed acyclic graph of dependencies (what depends on what).

**Dependency configurations** : named sets (groups) of dependencies (e.g. per task).

# Plugins

A plugin applies a set of extensions to the build process.

- ▶ Add tasks to a project.
- ▶ Pre-configure these tasks with reasonable defaults.
- ▶ Add dependency configurations.
- ▶ Add new properties and methods to existing objects.

Plugins implement the “build-by-convention” principle in a flexible way.

## Standard Gradle Plugins

## Gradle build language

- ▶ Gradle build scripts are Groovy scripts.
- ▶ The scripts operate in the context of Gradle's domain objects (DSL objects, aka delegates).
- ▶ The main DSL object is a `org.gradle.api.Project` instance.
- ▶ There is one-to-one relationship between a Project object and a `build.gradle` file.
- ▶ Each DSL object exposes its own properties and methods.

Have a look at: [Gradle DSL Reference](#)



# Back to the example

## Clarification

```
1 //a groovy script with a Project object as the context (delegate
  )
2 apply plugin: "java"
3 //invocation of Project.apply(Map) method
4 group    = "org.foo.something"
5 version  = "1.0-SNAPSHOT"
6 //update of project.group, project.version properties
7 repositories {
8     mavenCentral()
9 }
10 //invocation of the Project.repositories(Closure) method
11 //From the docs: the closure has a RepositoryHandler object
12 //as its delegate
13 dependencies {
14     compile    "commons-io:commons-io:2.4"
15     testCompile "junit:junit:4.11"
16     runtime    files("lib/foo.jar", "lib/bar.jar")
17 }
18 //here?
```

# A more complex example

```
1  apply plugin:'java'
2  //introduce a new task (that invokes a java process)
3  task generateFiles(type: JavaExec) {
4      main = 'some.class.name'
5      classpath = sourceSets.main.runtimeClasspath
6      args = [ projectDir, 'path/to/gen/files' ]
7  }
8  //uses the task 'keyword' (a special case) that
9  //ends up invoking one of the Project.task() methods
10 test { //customize the test task
11     dependsOn generateFiles
12     doLast {
13         ant.copy(toDir:'build/test-classes'){
14             fileset dir:'path/to/gen/files'
15         }
16     }
17 }
18 //import a class required by this build script
19 //some necessary classpath definitions are not shown for brevity
20 import org.apache.commons.io.FileUtils
21 clean.doFirst { //customize the clean task
22     FileUtils.deleteQuietly(new File('path/to/gen/files'))
23 }
```

Let's build a swing application

Thank you

